18a Differential equation for the components

Copy the kernel relations from 13b

In[ ]:=
```
Clear[h, p, t, f, r3m1, rm3m1]
r3m1[rr_] = {f[rr, t] → 1/(4 betac π t) ⅈ ((-2 + h - 2 p - rr) f[-2 + rr, t] + 2 t f^(0,1)[-2 + rr, t])}
rm3m1[rr_] = {f[rr, t] → 1/(4 beta π t) ⅈ ((2 + h + 2 p - rr) f[2 + rr, t] - 2 t f^(0,1)[2 + rr, t])}
```

Out[ ]=
$$\left\{f[rr,\ t] \to \frac{i\,((-2 + h - 2\,p - rr)\,f[-2 + rr,\ t] + 2\,t\,f^{(0,1)}[-2 + rr,\ t])}{4\ \text{betac}\ \pi\ t}\right\}$$

Out[ ]=
$$\left\{f[rr,\ t] \to \frac{i\,((2 + h + 2\,p - rr)\,f[2 + rr,\ t] - 2\,t\,f^{(0,1)}[2 + rr,\ t])}{4\ \text{beta}\ \pi\ t}\right\}$$

Take r such that -p ≤r≤p-2. Combine both relations for f_r and f_{r+2} to a second order differential equation for f_r.

In[ ]:=
```
frp = f[r + 2, t] /. r3m1[r + 2]
```

Out[ ]=
$$\frac{i\,((-4 + h - 2\,p - r)\,f[r,\ t] + 2\,t\,f^{(0,1)}[r,\ t])}{4\ \text{betac}\ \pi\ t}$$

In[ ]:=
```
Clear[r]
frp = f[r + 2, t] /. r3m1[r + 2](* f[r+2,t] expressed in
 f[r,t] and derivative by the first downward kernel relation *)
f[r, t] /. rm3m1[r](* f[r,t] expressed in f[r+2,t] and
 derivative by the second downward kernel relation *)
% - f[r, t](* resulting relation *) /. f[2 + r, t] → frp /.
   f^(0,1)[2 + r, t] → D[frp, t] // Simplify
eq = % /. betac → Abs[beta]^2 / beta // Simplify
```

Out[ ]=
$$\frac{i\,((-4 + h - 2\,p - r)\,f[r,\ t] + 2\,t\,f^{(0,1)}[r,\ t])}{4\ \text{betac}\ \pi\ t}$$

Out[ ]=
$$\frac{i\,((2 + h + 2\,p - r)\,f[2 + r,\ t] - 2\,t\,f^{(0,1)}[2 + r,\ t])}{4\ \text{beta}\ \pi\ t}$$

Out[ ]=
$$(-((-16 + h^2 - 16\,p - 4\,p^2 - 2\,h\,r + r^2 + 16\ \text{beta betac}\ \pi^2\,t^2)\,f[r,\ t]) + 4\,t\,(-((3 + 2\,p)\,f^{(0,1)}[r,\ t]) + t\,f^{(0,2)}[r,\ t]))\,/\,(16\ \text{beta betac}\ \pi^2\,t^2)$$

Out[ ]=
$$(-((-16 + h^2 - 16\,p - 4\,p^2 - 2\,h\,r + r^2 + 16\,\pi^2\,t^2\,\text{Abs}[\text{beta}]^2)\,f[r,\ t]) + 4\,t\,(-((3 + 2\,p)\,f^{(0,1)}[r,\ t]) + t\,f^{(0,2)}[r,\ t]))\,/\,(16\,\pi^2\,t^2\,\text{Abs}[\text{beta}]^2)$$

Write $f_r(t) = t^{p+2}\,j_r(t)$

```
In[ ]:= Clear[j, tau]
eq /. f[r, t] → t^(2 + p) j[r, 2 Pi Abs[beta] t] /. f^(0,ee_)[r, t] :→
    D[t^(2 + p) j[r, 2 Pi Abs[beta] t], {t, ee}] /. t → tau / (2 Pi Abs[beta]) // Simplify
Solve[% == 0, j^(0,2)[r, tau]][[1]] // Simplify
```

Out[ ]= $4^{-2-p} \pi^{-2\times(1+p)} \text{tau}^p \text{Abs[beta]}^{-2-p} \left(-\left(\left(h^2 (2\pi)^p - 2^{1+p} h \pi^p r + (2\pi)^p r^2 + 2^{2+p} \pi^p \text{tau}^2\right) j[r, \text{tau}]\right) + 2^{2+p} \pi^p \text{tau} \left(j^{(0,1)}[r, \text{tau}] + \text{tau}\, j^{(0,2)}[r, \text{tau}]\right)\right)$

Out[ ]= $\left\{j^{(0,2)}[r, \text{tau}] \to \frac{\left(h^2 - 2 h r + r^2 + 4 \text{tau}^2\right) j[r, \text{tau}] - 4 \text{tau}\, j^{(0,1)}[r, \text{tau}]}{4 \text{tau}^2}\right\}$

This is the modified Bessel differential equation with parameter
v=(h-r)/2.